

paul bjørnum

Foreløbigt tryk
af programmeringsafsnit til:

DATA- LÆRE

Grundbog med opgaver

PAUL BJØRNUM

Foreløbigt tryk af
programmeringsafsnit
til:

DATA LÆRE

Grundbog
med opgaver

FORLAGET BOGIKA 1981

Mekanisk, fotografisk eller anden gengivelse af denne bog eller dele af den er ikke tilladt ifølge gældende dansk lov om ophavsret.

Alle rettigheder forbeholdes forfatteren.

Bogen kan bestilles hos:

Forlaget BOGIKA ApS.
Akacieparken 38, 7430 Ikast
Tlf. (07) 15 31 55

Tryk:
Tarm Bogtryk & Offset A/S

Edb-maskiner og den teknologiske udvikling

Edb- og edb-maskiner eller datamater, som er det rigtige navn for maskinerne, står for mange som noget mystisk, svært og utilgængeligt.

Når man tidligere tænkte på en datamat, var det en stor kompliceret maskine, som betjentes af specialister, prisen for en datamat kunne være flere millioner kroner.

For at udnytte så dyr en maskine, må der være tale om endog meget store mængder af data, der skal behandles, eller man må være mange om at dele en maskine.

De meget store datamater vil vi vende tilbage til på et senere tidspunkt.

Her vil vi beskæftige os med den seneste udvikling inden for datamatteknikken.

Man taler om den teknologiske eksplosion.

Det er rigtigt, at udviklingen de seneste år har haft en sådan fart, ikke kun med hensyn til hvad maskinerne kan, men også med hensyn til maskinernes pris, at det kan være svært at følge med.

Til sammenligning med denne udvikling kunne det være morsomt at se, hvordan en folkevogn ville tage sig ud, hvis bilindustrien havde udviklet sig med samme hastighed som datamat-teknikken.



Folkevogn, der er fulgt med den teknologiske udvikling.

Pris: kr. 13,50

Bensinforbrug: en milliliter pr. 10 km.

Tophastighed: 100.000 km i timen.

Levetid: 10.000 år.

Ja, det var sager.

Men lad os komme ned på jorden igen og se på den datamat, der er afløseren for faktureringsmaskinerne og bogføringsmaskinerne.

Den er resultatet af den førnævnte teknologiske udvikling.

Mikrodatamaten

Med faktureringsseksemplet som repræsentant for det administrative arbejde, kan vi opstille følgende krav til »den ideelle maskine«.


Den skal kunne:

- skrive
- regne
- opbevare data om kunderne (kundekartotek)
- opbevare data om varerne (varekartotek)
- arbejde hurtigt
- arbejde sikkert.

Alle ovenstående krav opfyldes af »mikrodatamaten«.

En mikrodatamat består af enhederne:

Tastatur



Tastaturet minder om det tastatur, vi kender fra en almindelig skrivemaskine blot udvidet med nogle få funktionstaster.

Tastaturet bruges, når data skal »ind« i maskinen (inddata).

Fjernsynsskærm



Mikrodatamaten kan ikke tale (endnu), så hvis »operatøren« skal have resultater eller andre data fra maskinen, kommer de frem på skærmen (uddata).

Centralenhed




Centralenheden er maskinens »hjerne«.

Det er her, der regnes, og det er her, data opbevares.

Det er også centralenheden, der sørger for, at de forskellige opgaver, der skal udføres, bliver udført i den rigtige rækkefølge (styring af opgaveforløb).

Skrivemaskinen (printeren)



Når maskinen skal »aflevere« resultater, kan det som nævnt ske på skærmen.

Til mange opgaver er det nødvendigt med en udskrift på papir, tænk blot på fakturaen.

Den skrivemaskine, der er tilsluttet mikrodatamaten, er en hurtigskriver, der kan skrive med hastigheder på flere tusind bogstaver i minuttet.

Her er vist en mikrodatamat klar til brug.



En mikrodatamat kan intet udføre, før den gennem et program får at vide, nøjagtigt hvad vi ønsker, den skal gøre.

Vi må punkt for punkt give maskinen besked på, hvad den skal udføre.

Eks.

Skriv kundens navn
Beregn antal $\times$ pris
Skriv resultatet.

De regler, mikrodatamaten arbejder efter, kaldes et program.

Arbejdet med at udarbejde et program kaldes at programmere.

Programmering

Selve ordet »program« kender vi fra dagligdagen.

Vi slår op i avisen for at se dagens:

- radio program
- TV-program

Her er TV-programmet for lørdag.

LØRDAG
14.00 (TH) FORBRUGERNE - OG HJÆLPEMIDLER FOR MILLIONER
14.45 (TH) »DET KAN BETALE SIG AT KLAGE
15.10 (TH) UNGE SPØR' - DOMMEREN OM RET OG URET
15.50 (S/H) CLUNY BROWN Amerikansk lystspil fra 1946. Cluny Brown er en ung pige med et lidenskabeligt talent for blik- kenslagerarbejde; en egenskab, som undertiden fører hendes liv i nye retninger.
17.30 EUROPAS SIDSTE VILDMARK
18.30 SÅ ER DER TEGNEFILM
19.00 2. SAL TIL HØJRE OG TIL VENSTRE
19.30 TV-AVISEN
19.50 NÆSTE UGES TV
20.10 ET SKUD I MØRKE Engelsk komedie fra 1964. Politiinspektør Clouseau sættes på en sag, hvor en nydelig kam- merpige mistænkes for mord. Clouseau tror på hendes uskyld og på sin egen ufejlbarlighed.
LØRDAG - LIGE NU
23.00 SKUM Dutch slipper ud af fængslet, mens Burt stadig forsøger at flygte fra rumskibet. Danny kommer hjem med en ny pige og stakkels Jodie er helt alene.
23.25 TV-AVISEN

Dette TV-program er en liste, der punkt for punkt fortæller, hvad der er i fjernsynet den pågældende lørdag.

Edb-programmet har mange lighedspunkter med TV-programmet.

Det består som TV-programmet af tidspunkter (en rækkefølge) og oplysninger om, hvad der skal ske (programsætninger).

Når mikrodatamaten skal udføre bestemte opgaver, må programmerne, der skal styre maskinen, skrives i et sprog, som maskinen kender og kan forstå.

De fleste mikrodatamater forstår flere forskellige programmeringssprog.

Det sprog, vi har valgt at arbejde med, hedder »COMAL«.

Programmeringssproget COMAL

Datamaten skriver tekst.

TIMEN ER GOD
LÆREREN ER DYGTIG.

Teksten er skrevet af en mikrodatamat, sandhedsværdien skal ikke kommenteres, men programmet, der har bevirket udskrivningen, er her:

0010 PRINT "TIMEN ER GOD"
0020 PRINT "LÆREREN ER DYGTIG."

Linjenr.	Programsætning
----------	----------------

Linjenumrene 10, 20 sikrer, at programsætningerne udføres på det rigtige sted i programmet.

Programsætningerne fortæller maskinen, hvad den skal udføre.

Den første programsætning består af COMAL-ordet PRINT og teksten »TIMEN ER GOD«.

Bemærk at teksten, der skal skrives, står mellem anførselstegn " " .

PRINT

PRINT-SÆTNINGEN UDSKRIVER DEN TEKST, DER STÅR
OPFØRT MELLEM ANFØRSELSTEGNENE " "

Øvelse 4.1

0015 PRINT "VI MÅ VIDERE"

Hvilket linjenummer har programlinjen? _____

Hvilket COMAL-ord begynder sætningen med? _____

Hvilken tekst skriver maskinen? _____

Øvelse 4.2

Skriv et program, der giver udskriften:

MINE DAMER OG HERRER,
DETTE ER MIT FØRSTE PROGRAM.

Øvelse 4.3

Skriv et program, der skriver teksten:

RISTET BRØD ER LET AT LAVE,
BLOT MAN VIL ERINDRE:
NÅR DET OSER, SKAL DET HAVE
TO MINUTTER MINDRE.

Øvelser der bedst udføres ved maskinen.

PS: VIGTIGT. Find tasten ESCAPE **[ESC]**; denne tast kan bruges, hvis noget går galt, f.eks. hvis noget går i kludder under indtastningen af en linje, tryk **[ESC]** og du kan starte forfra på linjen.

Indtast nedenstående programlinje (foranstående nuller skal ikke indtastes).

0010 PRINT "DET ER DEJLIGT VEJR"

Har du husket at bruge 0 og ikke bogstavet O i linjenummeret.

Når linjen er indtastet trykkes på tasten »RETURN«.

Hvis linjen er i orden, læser maskinen linjen og gemmer den i arbejdslageret, klar til brug når operatøren ønsker det.

RETURN

RETURN-TASTEN BRUGES TIL AT »FORTÆLLE« MASKINEN, AT OPERATØREN ER FÆRDIG MED EN INDTASTNING.

Skriv nu ordet RUN og tryk RETURN.

Hvad skete der? _____

Øvelse 4.4

Skriv et program, der får maskinen til at skrive:

FOLK MED LIDT MEKANISK SNILDE
MÅ DET LIGEFREM GÅ ILDE

Indtast programmet og skriv RUN (RETURN).

Kontroller, om programmet udfører opgaven.

RUN

COMAL-KOMMANDOEN RUN BRUGES, NÅR MASKINEN SKAL UDFØRE ET PROGRAM, DER STÅR I ARBEJDSLAGE-RET.

Skriv ordet LIST (RETURN).

Hvad skete der? _____

LIST

COMAL-KOMMANDOEN LIST BRUGES, NÅR ET PROGRAM FRA ARBEJDSLAGE-RET ØNSKES SKREVET PÅ SKÆRMEN.

Skriv NEW (RETURN), og derefter LIST (RETURN).

Hvad skete der? _____

NEW

COMAL-KOMMANDOEN NEW BRUGES, NÅR ET PROGRAM I ARBEJDSLAGE-RET SKAL SLETTES.

Øvelse 4.5

Skriv LOAD ØVELSE1 (RETURN)

Skriv LIST (RETURN)

Hvad skete der? _____

LOAD

COMAL-KOMMANDOEN LOAD OVERFØRER ET PROGRAM FRA DET YDRE LAGER (DISKETTEN) TIL ARBEJDSLAGE-RET.

Kontroller, at du har dette program stående på skærmen:

```
0010 PRINT "ROSER ER RØDE"  
0020 PRINT "VIOLE ER SORTE"  
0030 PRINT "JORDBÆR ER SØDE"  
0040 PRINT "OG DU LIGE SA"
```

Hvor mange programlinjer består programmet af? _____

Hvad udfører programlinje nr. 30? _____

Skriv RUN

Hvilken udskrift giver programmet?

Skriv LIST (RETURN)

Der er en fejl i rimet; i hvilken programlinje? _____

Hvordan skulle linjen have set ud?

Fejlrettelse

Opdages en fejl, når programmet er lagret i maskinen, hjælper det ikke med et viskelæder.

Det er nødvendigt at kende fremgangsmåden ved fejlrettelse.

Ret fejlen i linje 20 ved at indtaste den rigtige linje med det samme linjenummer.

Skriv LIST (RETURN).

Er den nye linje nu på plads? _____

Skriv RUN – er programudskriftet i orden? _____

Er der ikke skrevet ja ved begge spørgsmål så prøv igen.

FEJL- RETTELSE

INDTASTES EN PROGRAMLINJE MED SAMME LINJENR. SOM EN TIDLIGERE, SLETTES DEN »GAMLE« LINJE, OG DEN SIDST INDTASTEDE ER GÆLDENDE.

Øvelse 4.6

Skriv LOAD ØVELSE2 og derefter LIST

Kontrollér om programmet, der nu står på skærmen, er:

```
0010 PRINT "I ØSTEN STIGER SOLEN OP"  
0020 PRINT "GÅR OVER HAV OG BJERGETOP"  
0030 PRINT "GÅR OVER LAND OG BY"
```

Find fejlen i verset og skriv her, hvad der er galt:

Indtast den manglende verslinje, men brug linjenummeret 15.

Skriv LIST

Hvad er der sket i programmet? _____

De springende linjenumre (10, 20, 30) gør det muligt at tilføje »glemte« programlinjer til et allerede skrevet og indlæst program.

Skriv RENUM og derefter LIST

Hvad er der sket? _____

RENUM

COMAL-KOMMANDOEN RENUM BEVIRKER, AT PROGRAMLINJERNE I ARBEJDSLAGERET OMNUMMERES 10, 20, 30 osv.

Øvelse 4.7

Skriv LOAD ØVELSE3 og derefter LIST

Kontrollér om dette program står på skærmen.

```
0010 PRINT "VIL DU MED RETTE"  
0020 PRINT "HA RY SOM EN LÆRD"  
0030 PRINT "DEN SPREDER GULD PÅ SKY"  
0040 PRINT "SÅ SØG DET LETTE-"  
0050 PRINT "OG GØR DET SVÆRT"
```

Skriv her hvilken linje, der ikke hører hjemme i rimet.

Skriv DEL + linjenummeret på den forkerte linje.

Skriv LIST

Hvad er der sket? _____

DEL

COMAL-KOMMANDOEN DEL FJERNER LINJER FRA ET PROGRAM.

Skriv RENUM og derefter LIST

Hvad er der sket? _____

Øvelse 4.8

Junior-programmøren i SALGSLYST er blevet sat til at lave et program, der udskriver følgende adresse:

A/S KROP & SJÆL
ATT: HR. A. SKULDERBLAD
BAGDELSSTRÆDE 37
1732 FODMOSE

Her er juniors program.

```
10 PRINT "A/S KROP & SJÆL"  
20 PRINT "BAGDELSSTRÆDE 37"  
30 PRINT "1732 FODMOSE"
```

Hvilke 2 fejl har junior begået?

Hvordan kan fejlene rettes?

Indtast programmet og afprøv det ved at taste RUN.

Øvelse 4.9

Skriv LOAD ØVELSE4 og derefter LIST

Kontrollér programmet på skærmen med udskriften herunder.

```
0010 PRINT
0020 PRINT "SKRIVES PRINT SOM HER"
0030 PRINT
0040 PRINT "UDEN ORD, OG FOR SIG SELV"
0050 PRINT
0060 PRINT "FAR DU LUFT, SOM DU SER"
0070 PRINT
0080 PRINT "DET ER FINT PROGRAMMEL."
0090 PRINT
```

Tast RUN

Hvad bevirker de »tomme« PRINT-sætninger?

Øvelse 4.10

Skriv LOAD ØVELSE5 og derefter LIST

Kontrollér programmet med udskriften herunder.

```
0010 PRINT "NU ER DET FORÅR I LUFTEN"
0020 PRINT "EN LUNHED OG LYS OG LETHED"
0030 PRINT "OG VINTERTRÆTHEDEN ER GÆT OVER"
0040 PRINT "I FORÅRSTRÆTHED"
```

Ret programmet, så det starter og slutter med linjeskift, og der bliver dobbelt linjeafstand mellem linjerne.

Brug det system, du lærte i øvelserne 4.9.

Tast LIST

Tast RENUM og derefter LIST

Hvad er der sket med de nye linjer? _____

Tast RUN; er programmet i orden? _____

Læsning til VARIABLE

KÆRE TIPPER

DET GLÆDER OS AT KUNNE MEDDELE DEM,
AT DE HAR VUNDET KR. 5000 I TIPNING

MED VENLIG HILSEN
VESTJYSK TIPSTJENESTE

Brevet herover ønsker Vestjysk Tipstjeneste udskrevet automatisk fra mikrodatamaten.

Øvelse 4.11

Skriv programmet til Vestjysk Tipstjeneste.

Indtast og afprøv programmet.

Vestjysk Tipstjeneste er tilfreds med programmet den første uge, men allerede ugen efter går det galt; der er gevinsten 7000 kr.

VARIABLE

Gevinstens størrelse ændrer sig fra uge til uge, den er »variabel«.
Vi kunne istedet lade brevet se således ud:

KÆRE TIPPER

DET GLÆDER OS AT KUNNE MEDDELE DEM,
AT DE HAR VUNDET KR. <GEVINST> I TIPNING

MED VENLIG HILSEN
VESTJYSK TIPSTJENESTE

Det er i orden, hvis <GEVINST> har værdien »det aktuelle gevinstbeløb for den pågældende uge«.

Det problem klarer vi med programlinjen:

```
0010 INPUT GEVINST
```

Sætningen består af COMAL-ordet INPUT og variabelen GEVINST.

Når maskinen læser COMAL-ordet INPUT, gør den klar til at modtage indtastningen af en variabel.

I denne sætning har vi bestemt, at den variable skal hedde GEVINST.

Indtastes 5000, vil variabelen GEVINST få værdien 5000.

Arbejdslageret i datamaten kan sammenlignes med et reolsystem med tomme hylder:

Når sætningen INPUT GEVINST læses, reserveres en hylde i arbejdslageret, og den får navnet GEVINST.

GEVINST					

Indtastes tallet 5000, vil tallet blive lagt ind på hylden GEVINST.

GEVINST					
5000					

Programlinjen

0020 PRINT GEVINST

bevirker, at »indholdet« af hylden GEVINST udskrives.

Har GEVINST værdien 5000, vil maskinen skrive dette tal.

Programmet fra Vestjysk Tipstjeneste vil nu se sådan ud.

```

0010 INPUT GEVINST
0020 PRINT
0030 PRINT "KÆRE TIPPER"
0040 PRINT
0050 PRINT "DET GLÆDER OS AT KUNNE MEDDELE DEM,"
0060 PRINT "AT DE HAR VUNDET KR. ",GEVINST," I TIPNING"
0070 PRINT
0080 PRINT "
0090 PRINT "
MED VENLIG HILSEN"
VESTJYSK TIPSTJENESTE"

```

Øvelse 4.12 Hvad udfører linje 10? _____

Hvad udfører linje 20? _____

Skriv linje 60 på papiret og bemærk tegnfordelingen omkring den variable **GEVINST**.

INPUT INPUT-SÆTNINGER BRUGES, NÅR VÆRDIER SKAL IND-
TASTES TIL VARIABLE.

TASTES TIL VARIABLE.

Øvelse 4.13

```
0010 PRINT "GEVINST"
0020 PRINT GEVINST
```

PRINT-SÆTNINGEN BRUGES, NÅR INDHOLDET AF EN VARIABEL SKAL UDSKRIVES.

Øvelse 4.14 Skriv LOAD ØVELSE6 og derefter LIST.

Kontrollér programmet på skærmen med udskriftet herunder.

```
0010 INPUT GEVINST
0020 PRINT
0030 PRINT "KÆRE TIPPER"
0040 PRINT
0050 PRINT "DET GLÆDER OS AT KUNNE MEDDELE DEM,"
0060 PRINT "AT DE HAR VUNDET KR. ",GEVINST," I TIPNING"
0070 PRINT
0080 PRINT "                                MED VENLIG HILSEN"
0090 PRINT "                                VESTJYSK TIPSTJENESTE"
```

Skriv RUN.

Når maskinen »venter«, er det for at få indtastet den variable.

Prøv igen med nye gevinstbeløb, husk at taste RUN.

Øvelse 4.15 Hotel »Gitterly« anvender dette program til udskrivning af breve til »gæsterne«.

```

0010 PRINT "INDTAST VÆRELSE NUMMER"
0020 INPUT VÆRELSENØR
0030 PRINT "INDTAST BELØB"
0040 INPUT BELØB
0050 PRINT
0060 PRINT "KÆRE GÆST"
0070 PRINT
0080 PRINT "VI HAR GENNEMGÅET DERES VÆRELSE NR ",VÆRELSENØR
0090 PRINT "DER MANGLER INVENTAR FOR KR. ",BELØB
0100 PRINT "BELØBET BEDES INDBETALT MÆRKET VÆRELSE ",VÆRELSENØR
0110 PRINT
0120 PRINT "                                MED VENLIG HILSEN"
0130 PRINT "                                VÆRTEN"

```

Hvor mange forskellige variable indeholder programmet?

Hvad hedder de variable?

I hvilke programlinjer udskrives variable?

Hvad udfører linje 10? _____

Hvad udfører linje 20? _____

Hvad udfører linje 30? _____

Hvad udfører linje 40? _____

Hvad udfører linje 70? _____

Øvelse 4.16

LOAD ØVELSE7

Skriv LIST og kontrollér, om det er programmet fra hotel »GITTERLY«.

```
0010 PRINT "INDTAST VÆRELSE NUMMER"
0020 INPUT VÆRELSEN
0030 PRINT "INDTAST BELØB"
0040 INPUT BELØB
0050 PRINT
0060 PRINT "KÆRE GÆST"
0070 PRINT
0080 PRINT "VI HAR GENNEMGÅET DERES VÆRELSE NR ",VÆRELSEN
0090 PRINT "DER MANGLER INVENTAR FOR KR. ",BELØB
0100 PRINT "BELØBET BEDES INDBETALT MÆRKET VÆRELSE ",VÆRELSEN
0110 PRINT
0120 PRINT "                                MED VENLIG HILSEN"
0130 PRINT "                                VÆRTEN"
0010 PRINT "INDTAST VÆRELSE NUMMER",
0030 PRINT "INDTAST BELØB",
```

Kør programmet (skriv RUN), brug variabelværdierne 23 og 112.50 (bemærk punktum i stedet for komma i beløbet).

Er brevet i orden?

Prøv selv med andre værdier for værelse nummer og beløb. (Husk RUN).

Øvelse 4.17

Skriv LIST og indtast derefter linjerne 10 og 30 således:

```
0010 PRINT "INDTAST VÆRELSE NUMMER",  
0030 PRINT "INDTAST BELØB",
```

Skriv RUN

Hvad skete der? _____

PRINT,

KOMMA EFTER TEKST I EN PRINT-SÆTNING BEVIRKER, AT MASKINEN MODTAGER EN EFTERFØLGENDE INPUT-SÆTNING PÅ SAMME LINJE.

Øvelse 4.18

Hvad udfører nedenstående program?

```
0010 INPUT HØJDE  
0020 INPUT VÆGT  
0030 INPUT ALDER  
0040 PRINT HØJDE  
0050 PRINT VÆGT  
0060 PRINT ALDER
```

Skriv LOAD ØVELSE8 og derefter LIST.

Kontrollér, om det er ovenstående program, der står på skærmen.

Skriv RUN og indtast 180 (return), 76 (return) og 21 (return).

Hvad udførte programmet? _____

**KONVERSE-
RENDE FORM**

Indtast nu følgende programlinjer:

```
0005 PRINT "INDTAST HØJDE ",  
0015 PRINT "INDTAST VÆGT ",  
0025 PRINT "INDTAST ALDER ",  
0035 PRINT  
0040 PRINT "HØJDEN ER ", HØJDE  
0050 PRINT "VÆGTEN ER ", VÆGT  
0060 PRINT "ALDEREN ER ", ALDER
```

Skriv LIST – er linjerne på plads?

Skriv RENUM og derefter LIST.

Er linjerne nu nummereret 10, 20, 30 osv.?

Tast RUN.

Hvilke fordele er der ved det nye program i forhold til programmet, inden det blev rettet?

**PROGRAM-
MØREN SKAL
ANVENDE
KONVERSE-
RENDE FORM**

NÅR ET PROGRAM ANVENDER INPUTSÆTNINGER, SKAL PROGRAMMET INDEHOLDE »ANVISNINGER« TIL OPERATØREN; MAN KALDER ARBEJDET SAMMEN MED MASKINEN FOR KONVERSERENDE.

Øvelse 4.19

Skriv et program, hvor maskinen på skærmen skriver:

INDTAST ET ÅRSTAL.

Derefter skal maskinen skrive:

ÅRSTALLET ER xxxx.

Husk at bruge INPUT-sætning til ÅRSTAL.

Indtast og afprøv programmet med forskellige årstal.

Øvelse 4.20

Skriv et program, der giver følgende udskrift:

DAGENS MIDDAG KOSTER xxx.xx.

Husk at bruge den konverserende form.

Indtast og afprøv programmet med forskellige priser.

Øvelse 4.21

Skriv et program, der udskriver nedenstående rykkerbrev, idet kontonummer og beløb skal være variable. (Husk konverserende form under modtagelsen af de variable data).

VED GENNEMGANG AF VORT BOGHOLDERI
ERFARER VI, AT DERES KONTO NR. XXXX
UDVISER EN SALDO I VOR FAVØR PÅ KR. XXXX.XX

Indtast og afprøv programmet med forskellige data for de variable.

Øvelse 4.22

FRUGTLAGERET A/S
BANANVEJEN 37
3098 URSKOVBY

DERES ORDRE NR. 10512

I HENHOLD TIL OVENNÆVNTE ORDRE HAR VI I DAG
AFSENDT:

15 KASSER APPELSINER

VI VEDLÆGGER FAKTURA NR. 32416 PÅ KR. 1890,00
BELØBET ER DEBITERET DERES KONTO

MED VENLIG HILSEN
DANSK APPELSINIMPORT

Ovennævnte brev er skrevet automatisk af en mikrodatamat.

Programmøren havde endnu ikke lært om variable og konver-
rende form, så hans program ser således ud:

```
0010 PRINT "FRUGTLAGERET A/S"
0020 PRINT "BANANVEJEN 37"
0030 PRINT "3098  URSKOVBY"
0040 PRINT
0050 PRINT "DERES ORDRE NR. 10512"
0060 PRINT
0070 PRINT "I HENHOLD TIL OVENNÆVNTE ORDRE HAR VI I DAG"
0080 PRINT "AFSENDT:"
0090 PRINT
0100 PRINT "15 KASSER APPELSINER"
0110 PRINT
0120 PRINT "VI VEDLÆGGER FAKTURA NR. 32416 PÅ KR. 1890,00"
0130 PRINT "BELØBET ER DEBITERET DERES KONTO"
0140 PRINT
0150 PRINT "MED VENLIG HILSEN"
0160 PRINT "DANSK APPELSINIMPORT"
```

De variable, vi først og fremmest ønsker mulighed for at kunne
indtaste, er her indrammede.

FRUGTLAGERET A/S
BANANVEJEN 37
3098 URSKOVBY

DERES ORDRE NR. 10512

I HENHOLD TIL OVENNÆVNTE ORDRE HAR VI I DAG
AFSENDT:

15 KASSER APPELSINER

VI VEDLÆGGER FAKTURA NR. 32416 PÅ KR. 1890,00
BELØBET ER DEBITERET DERES KONTO

MED VENLIG HILSEN
DANSK APPELSINIMPORT

Anvend følgende navne til de variable:

ORDRENUMMER
ANTAL
FAKTURANUMMER
BELØB

Skriv de linjer, der skal tilføjes programmet.

og ret de nødvendige linjer:

Det oprindelige program kan kaldes frem fra disketten ved at skrive LOAD ØVELSE9.

Kontrollér, om det er dette program, der står på skærmen.

```
0010 PRINT "FRUGTLAGERET A/S"
0020 PRINT "BANANVEJEN 37"
0030 PRINT "3098  URSKOVBY"
0040 PRINT
0050 PRINT "DERES ORDRE NR. 10512"
0060 PRINT
0070 PRINT "I HENHOLD TIL OVENNÆVNTE ORDRE HAR VI I DAG"
0080 PRINT "AFSENDT:"
0090 PRINT
0100 PRINT "15 KASSER APPELSINER"
0110 PRINT
0120 PRINT "VI VEDLÆGGER FAKTURA NR. 32416 PÅ KR. 1890,00"
0130 PRINT "BELØBET ER DEBITERET DERES KONTO"
0140 PRINT
0150 PRINT "MED VENLIG HILSEN"
0160 PRINT "DANSK APPELSINIMPORT"
```

Indfør de nye linjer og afprøv programmet med forskellige variable data.

Til sammenligning med dit eget rettede program kan du skrive LOAD ØVELSE10.

Datamaten udfører regnefunktioner.

Øvelse 4.23

Læs nedenstående program.

```
0010 TIMEANTAL:=8
0020 TIMELØN:=40
0030 DAGLØN:=TIMEANTAL*TIMELØN
0040 PRINT "DAGLØNNEN ER KR. ",DAGLØN
```

Lad os se hvad linje 10 udfører.

```
0010 TIMEANTAL:=8
```

DYNAMISK LIGHEDSTEGN

Bemærk det specielle lighedstegn »:=« det kaldes et dynamisk lighedstegn, og det læses »sættes til at være lig med«.

Linjen kan nu læses således:

TIMEANTAL sættes til at være lig med 8.

Det betyder, at variabelen »TIMEANTAL« får tildelt værdien 8.

Linje 20.

```
0020 TIMELØN:=40
```

Hvordan skal linjen læses?

Hvilken værdi har variabelen »TIMELØN«, når linjen er udført?

Linje 30.

```
0030 DAGLØN:=TIMEANTAL*TIMELØN
```

I denne linje får den nye variabel »DAGLØN« tildelt værdien TIMEANTAL × TIMELØN (bemærk at gangetegnet i COMAL er en stjerne *).

Hvilken værdi har variabelen »DAGLØN«, når linje 30 er udført?

Linje 40.

```
0040 PRINT "DAGLØNNEN ER KR. ",DAGLØN
```

Linjen bevirker udskriften:

```
DAGLØNNEN ER KR. 320
```

Indtast og afprøv programmet. (Bemærk at man under indtastningen ikke behøver at indtaste lighedstegnet som et dynamisk lighedstegn »:=«, men kun behøver at indtaste et almindeligt lighedstegn »=«; maskinen vil selv sørge for at omdanne dette til det dynamiske).

Øvelse 4.24

Lønprogrammet ønskes ændret, så der kan indtastes variabelt »TIMEANTAL« og »TIMELØN«.

Hvordan kommer programmet til at se ud?

Indtast og afprøv programmet med forskellige data.

Hvis dit program ikke virker, kan du taste LOAD ØVELSE11 og sammenligne dit program med det lagrede lønprogram.

```
0010 PRINT "INDTAST TIMEANTAL ",
0020 INPUT TIMEANTAL
0030 PRINT "INDTAST TIMELØN ",
0040 INPUT TIMELØN
0050 DAGLØN:=TIMEANTAL*TIMELØN
0060 PRINT "DAGLØNNEN ER KR. ",DAGLØN
```

Øvelse 4.25

Læs nedenstående program:

```
0010 ANTAL:=15
0020 PRIS:=25
0030 IALT:=ANTAL*PRIS
0040 PRINT "PRIS IALT ER KR. ",IALT
```

Hvad bevirker linje 10? _____

Hvad bevirker linje 20? _____

Hvad bevirker linje 30? _____

Hvad udskriver linje 40? _____

Indtast og afprøv programmet.

Øvelse 4.26

Læs nedenstående program:

```
0010 ANTAL:=50
0020 PRIS:=15
0030 RABATSATS:=20
0040 BRUTTOBELØB:=ANTAL*PRIS
0050 RABAT:=BRUTTOBELØB*20/100
0060 NETTOBELØB:=BRUTTOBELØB-RABAT
0070 PRINT "BRUTTOBELØBET ER KR.  ",BRUTTOBELØB
0080 PRINT "RABATBELØBET ER KR.   ",RABAT
0090 PRINT "NETTOBELØBET ER KR.   ",NETTOBELØB
```

Hvor mange variable benytter programmet? _____

Hvad hedder de variable, som programmet anvender?

Hvilken værdi har BRUTTOBELØB efter udførelsen af linje 40?

Hvilken værdi har RABAT efter udførelsen af linje 50?
(Bemærk: / betyder i COMAL division).

Hvilken værdi har NETTOBELØB efter udførelsen af linje 60?

Øvelse 4.27

Tast LOAD ØVELSE12 og kontrollér, om det er det rigtige program.

Afprøv programmet ved at skrive RUN.

Programmet skal nu skrive:

```
BRUTTOBELØBET ER KR.  750
RABATBELØBET ER KR.   150
NETTOBELØBET ER KR.   600
```

Programmet ønskes ændret, så der kan indtastes variabelt »ANTAL«, »PRIS« og »RABATSATS«. Husk konverserende form.

[illegible]

Øvelse 4.28

Indtast nu følgende til de variable:

ANTAL: 89
PRIS: 11.79
RABATSATS: 27

Hvilken udskrift giver maskinen?

Hvis det er gået, som det skal, er dit svar:

BRUTTOBELØBET ER KR. 1049.31
RABATBELØBET ER KR. 283.3137
NETTOBELØBET ER KR. 765.9963

Dette resultat er rigtigt, men det er ikke kønt.

Maskinen kan ikke vide, at det er et kronebeløb, du ønsker, den skal skrive.

Prøv nu at indtaste nedenstående sætninger:

PRINT-USING

```
0100 PRINT "BRUTTOBELØBET ER KR. ",
0110 PRINT USING "#####.##": BRUTTOBELØB
0120 PRINT "RABATBELØBET ER KR. ",
0130 PRINT USING "#####.##": RABAT
0140 PRINT "NETTOBELØBET ER KR. ",
0150 PRINT USING "#####.##": NETTOBELØB
```

Tast LIST og sammenlign med programmet herunder.

```
0010 PRINT "INDTAST ANTAL      ",
0020 INPUT ANTAL
0030 PRINT "INDTAST PRIS      ",
0040 INPUT PRIS
0050 PRINT "INDTAST RABATSATS ",
0060 INPUT RABATSATS
0070 BRUTTOBELØB:=ANTAL*PRIS
0080 RABAT:=BRUTTOBELØB*RABATSATS/100
0090 NETTOBELØB:=BRUTTOBELØB-RABAT
0091 SELECT OUTPUT "LP:"
0100 PRINT "BRUTTOBELØBET ER KR. ",
0110 PRINT USING "#####.##": BRUTTOBELØB
0120 PRINT "RABATBELØBET ER KR. ",
0130 PRINT USING "#####.##": RABAT
0140 PRINT "NETTOBELØBET ER KR. ",
0150 PRINT USING "#####.##": NETTOBELØB
```

Skriv RUN og indtast igen nedenstående til de variable.

ANTAL: 89
PRIS: 11.79
RABATSATS: 27

Hvilken udskrift står der nu på skærmen?

Var det denne?

BRUTTOBELØBET ER KR. 1049.31
RABATBELØBET ER KR. 283.31
NETTOBELØBET ER KR. 766.00

Hvilken virkning har linje 110, 130 og 150 haft på udskriften?

Hvad er der blevet af de »gamle« linjer 100, 110, 120?

Lad os se på den »gamle« udskrift af »NETTOBELØB«.

0120 PRINT "NETTOBELØBET ER KR. ", NETTOBELØB

Og på den nye:

0140 PRINT "NETTOBELØBET ER KR. ",
0150 PRINT USING "#####.##": NETTOBELØB

Noter forskellen herunder:

PRINT USING

VARIABLE KAN UDSKRIVES EFTER ET BESTEMT MØNSTER VED AT ANVENDE COMAL-SÆTNINGEN »PRINT USING« EFTERFULGT AF FORMATET PÅ DET ØNSKEDE UDSKRIFT.

EKS.: PRINT USING "#####.##": BELØB

VIL BEVIRKE, AT VARIABLEN »BELØB« UDSKRIVES MED 5 PLADSER FORAN KOMMAET OG 2 PLADSER EFTER KOMMAET.

Øvelse 4.29

Hvilken udskrift giver nedenstående program?

```
0010 ANTAL:=16.879
0020 PRIS:=19.84
0030 BELØB:=ANTAL*PRIS
0040 PRINT "BELØBET ER ",BELØB
```

Ret programmet så BELØB bliver skrevet med 5.2 cifre.

Ser dit program således ud?

```
0010 ANTAL:=16.879
0020 PRIS:=19.84
0030 BELØB:=ANTAL*PRIS
0040 PRINT "BELØBET ER ",
0050 PRINT USING "#####.##": BELØB
```

Indtast og afprøv programmet.

Øvelse 4.30

Skriv et program, hvor du ved at indtaste beløbet 1235.79 kan få nedenstående udskrift.

```
BELØBET EXCL. MOMS      1235.79
MOMSEN UDGØR            271.87
BELØBET INCL. MOMS      1507.66
```

Brug følgende variabelnavne:

NETTOBELØB
MOMS
BRUTTOBELØB

MOMS beregnes ved at skrive programlinjen:

```
0030 MOMS:=NETTOBELØB*22/100
```

BRUTTOBELØB beregnes ved at skrive programlinjen:

```
0040 BRUTTOBELØB:=NETTOBELØB+MOMS
```

Indtast og afprøv programmet med andre tal for NETTOBELØB.

```

0010 PRINT "INDTAST BELØB EXCL. MOMS ",
0020 INPUT NETTOBELØB
0030 MOMS:=NETTOBELØB*22/100
0040 BRUTTOBELØB:=NETTOBELØB+MOMS
0050 PRINT "BELØBET EXCL. MOMS ",
0060 PRINT USING "#####.##": NETTOBELØB
0070 PRINT "MOMSEN UDGØR           ",
0080 PRINT USING "#####.##": MOMS
0090 PRINT "BELØBET INCL. MOMS ",
0100 PRINT USING "#####.##": BRUTTOBELØB

```

Skriv et program, hvor du ved at indtaste beløbet 1295.98 kan få nedenstående udskrift:

BEØBET INCL. MOMS	1295.98
MOMSEN UDØR	233.67
BEØBET EXCL. MOMS	1062.31

Momsen udregnes ved at anvende programlinjen:

0030 MOMS:=BRUTTOBEL0B*18.03/100

This image shows a single sheet of white paper with horizontal blue or grey ruling lines. The lines are evenly spaced and run across the width of the page. There are approximately 20 lines visible. The paper has a slightly aged or off-white appearance. There is no handwriting or other markings on the page.

```
0010 PRINT "INDTAST BELØB INCL. MOMS ",
0020 INPUT BRUTTOBELØB
0030 MOMS:=BRUTTOBELØB*18.03/100
0040 NETTOBELØB:=BRUTTOBELØB-MOMS
0050 PRINT "BELØBET INCL. MOMS ",
0060 PRINT USING "#####.##": BRUTTOBELØB
0070 PRINT "MOMSEN UDGØR          ",
0080 PRINT USING "#####.##": MOMS
0090 PRINT "BELØBET EXCL. MOMS ",
0100 PRINT USING "#####.##": NETTOBELØB
```

54

Som det fremgår af nedenstående fire programlinjer, der alle stammer fra nogle af vore tidligere programmer, har vi nu været gennem alle 4 regnearter.

0040 BRUTTOBELØB:=NETTOBELØB+MOMS	+
0040 NETTOBELØB:=BRUTTOBELØB-MOMS	-
0030 BELØB:=ANTAL*PRIS	*
0080 RABAT:=BRUTTOBELØB*20/100	/

REGNETEGN I COMAL	1. addition der i COMAL udtrykkes som	+
	2. subtraktion der udtrykkes som	-
	3. multiplikation der udtrykkes som	*
	4. division der udtrykkes som	/

Det er muligt at bruge parenteser efter samme regler som dem, vi kender fra matematikken.

Øvelse 4.32

0010 FORÅRSVEJR:=SOL+(SOMMERSOL*(VEJR+TØ/18))

I hvilken rækkefølge vil regneoperationerne blive udført i ovenstående programlinje?

Maskinen opsummerer tal.

Hvad viser denne udskrift?

NETTOBELØBET ER KR.	3874.11
+ MOMS 22%	852.30
BRUTTOBELØBET ER KR.	4726.41

Hvad du ikke kan se af udskriftet er, at NETTOBELØB er en sum af 3 linjebeløb:

LINJEBELØB 1	der er indtastet med	100.50
LINJEBELØB 2	der er indtastet med	3.517.45
LINJEBELØB 3	der er indtastet med	256.16

Her er programmet der har bevirket udskriften:

```
0010 NETTOBELØB:=0
0020 PRINT "INDTAST 1. LINJEBELØB",
0030 INPUT LINJEBELØB
0040 NETTOBELØB:=NETTOBELØB+LINJEBELØB
0050 PRINT "INDTAST 2. LINJEBELØB",
0060 INPUT LINJEBELØB
0070 NETTOBELØB:=NETTOBELØB+LINJEBELØB
0080 PRINT "INDTAST 3. LINJEBELØB",
0090 INPUT LINJEBELØB
0100 NETTOBELØB:=NETTOBELØB+LINJEBELØB
0110 PRINT "NETTOBELØBET ER KR.      ",
0120 PRINT USING "#####.##": NETTOBELØB
0130 MOMS:=NETTOBELØB*22/100
0140 PRINT "+ MOMS 22%                ",
0150 PRINT USING "#####.##": MOMS
0160 BRUTTOBELØB:=NETTOBELØB+MOMS
0170 PRINT "BRUTTOBELØBET ER KR.      ",
0180 PRINT USING "#####.##": BRUTTOBELØB
```

Øvelse 4.33

Hvilke variable er brugt i programmet?

Hvad udfører linje 10? _____

REGEL

NÅR EN VARIABEL BRUGES TIL OPSUMMERING, SKAL VARIABLEN ALTID NULSTILLES, INDEN OPSUMMERINGEN PÅBEGYNDES.

Hvad udfører linje 100? _____

Hvad udfører linje 130? _____

Hvad udfører linje 160? _____

Linje 100 skal læses således:

NETTOBELØB sættes til at være lig med sig selv (det, den var i forvejen) + LINJEBELØB.

Det er altså i linjerne 40, 70 og 100, vi opsummerer NETTOBELØBET.

Øvelse 4.34

Tast LOAD ØVELSE14, list programmet og sammenlign med det ovenstående program.

Afprøv programmet, først med de nævnte 3 LINJEBELØB, derefter med nogle linjebeløb, du selv finder på.

Øvelse 4.35

Skriv et program, hvor du indtaster 2 tal og opsummerer dem.
Programmet skal give nedenstående udskrift:

SUMMEN AF DE 2 TAL ER: XXXXX.XX

Indtast og afprøv dit program.

Øvelse 4.36

Skriv et program, hvor du indtaster et timeantal og en timeløn, 3 gange.

DAGLØN = TIMEANTAL × TIMELØN.

LØN IALT = SUMMEN AF DAGLØN.

Programmet skal give nedenstående udskrift:

TIMEANTALLET IALT ER XXXX
UDBETALT LØN IALT UDGØR KR. XXXX.XX

Indtast og afprøv dit program.

Maskinen træffer logiske valg.

Øvelse 4.37

Læs nedenstående program grundigt igennem:

```
0010 PRINT "INDTAST TALLET 1 ELLER 2",
0020 INPUT TAL
0030 PRINT
0040 IF TAL=1 THEN
0050 PRINT "DU HAR VALGT TALLET 1 GAMLE VEN"
0060 ENDIF
0070 IF TAL=2 THEN
0080 PRINT "DU HAR VALGT TALLET 2 SMUKKE VEN"
0090 ENDIF
0100 IF TAL>2 THEN
0110 PRINT "HVORDAN VIL DU LÆRE PROGRAMMERING,"
0120 PRINT "NÅR DU IKKE ENGANG KAN INDTASTE DET"
0130 PRINT "RIGTIGE TAL, FJOLS"
0140 ENDIF
```

Hvilke variable bruger programmet? _____

IF-THEN

Lad os se på linjerne 40 og 60.

```
0040 IF TAL=1 THEN
0060 ENDIF
```

Linje 40 kan oversættes til:

HVIS TAL = 1 SÅ

og linje 60 kunne oversættes til:

SLUT PÅ HVIS

Ser vi forløbet af linjerne 40, 50, og 60

```
0040 IF TAL=1 THEN
0050 PRINT "DU HAR VALGT TALLET 1 GAMLE VEN"
0060 ENDIF
```

kan linjerne oversættes til:

HVIS TAL = 1 SÅ SKRIV, DU HAR VALGT TALLET 1, GAMLE
VEN.

REGEL

IF-THEN SÆTNINGEN BEVIRKER, AT PROGRAMMET, DER
ER ANFØRT MELLEM IF-THEN OG ENDIF UDFØRES, HVIS
IF-THEN SÆTNINGEN ER SAND. (I eksemplet hvis TAL = 1).

HVIS SÆTNINGEN DERIMOD ER FALSK, (hvis TAL ikke er =
1) SPRINGER MASKINEN OVER DE PROGRAMLINJER, DER
STÅR MELLEM IF-THEN OG ENDIF.

Øvelse 4.38

Hvad sker der i linjerne 70, 80 og 90?

70 _____

80 _____

90 _____

Tegnet > betyder større end.

Oversæt hvad der står i linje 100.

Øvelse 4.39

Tast LOAD ØVELSE15, kontrollér at du har fået det rigtige program, og prøv programmet ved at indtaste forskellige tal til den variable.

**RELATIONS-
OPERATORER**

Tegnene =, <, > osv. kaldes »relationsoperatorer«.

I COMAL ER FØLGENDE RELATIONSOPERATORER TILLADT I IF-THEN SÆTNINGER.

TEGN	BETYDNING	EKSEMPEL	I COMAL
>	større end	alder > 18 år	>
<	mindre end	alder < 18 år	<
=	lig med	alder = 18 år	=
≥	større end eller lig med	alder ≥ 18 år	>=
≤	mindre end eller lig med	alder ≤ 18 år	<=
≠	forskellig fra	alder ≠ 18 år	<>

Øvelse 4.40

Skriv et program, hvor du indtaster alderen på en person; hvis alderen er »mindre end 18« skal maskinen skrive:

PERSONEN ER IKKE PERSONLIG MYNDIG

Er alderen »større end eller lig med 18«, skal maskinen skrive:

PERSONEN ER PERSONLIG MYNDIG

Indtast og afprøv dit program.

Ønsker du at sammenligne dit program med det lagrede, kan du taste LOAD ØVELSE16.

```
0010 PRINT "INDTAST PERSONENS ALDER",
0020 INPUT ALDER
0030 IF ALDER<18 THEN
0040 PRINT
0050 PRINT "PERSONEN ER IKKE PERSONLIG MYNDIG"
0060 ENDIF
0070 IF ALDER>=18 THEN
0080 PRINT
0090 PRINT "PERSONEN ER PERSONLIG MYNDIG"
0100 ENDIF
```

Øvelse 4.41

Læs nedenstående program igennem.

```
0010 PRINT "INDTAST PERSONENS ALDER",
0020 INPUT ALDER
0030 IF ALDER>=18 THEN
0040   PRINT
0050   PRINT "PERSONEN ER PERSONLIG MYNDIG"
0060 ELSE
0070   PRINT
0080   PRINT "PERSONEN ER IKKE PERSONLIG MYNDIG"
0090 ENDIF
```

Hvilke forskelle er der mellem dette program og programmet i øvelse 4.40?

Lad os se på linje 60:

```
0060 ELSE
```

ELSE kan oversættes til ELLERS.

Forløbet i linjerne 30 til 90 kan nu læses således:

```
HVIS ALDER er større end eller lig med 18 SÅ skriv
PERSONEN ER PERSONLIG MYNDIG
ELLERS
skriv PERSONEN ER IKKE PERSONLIG MYNDIG.
```

ELSE

PROGRAMSÆTNINGEN ELSE BEVIRKER, AT MASKINEN KAN VÆLGE MELLEM TO MULIGHEDER I EN IF-THEN SÆTNING.

MASKINEN VÆLGER DEN FØRSTE MULIGHED, HVIS IF-THEN ER SAND, ELLERS VÆLGER MASKINEN ANDEN MULIGHED.

Øvelse 4.42

Skriv et program, der udskriver:

DET ER DEJLIGT VEJR

hvis du indtaster vejrtype 1, og

DET ER DÅRLIGT VEJR

hvis du indtaster et hvert andet tal end 1.

Øvelse 4.43

Læs nedenstående program:

```
0010 TIMERIALT:=0
0020 LØNIALT:=0
0030 PRINT "INDTAST ANTAL TIMER          ",
0040 INPUT TIMEANTAL
0050 PRINT "INTAST TIMELØNNEN          ",
0060 INPUT TIMELØN
0070 TIMERIALT:=TIMERIALT+TIMEANTAL
0080 DAGLØN:=TIMEANTAL*TIMELØN
0090 LØNIALT:=LØNIALT+DAGLØN
0100 PRINT
0110 PRINT "TIMEANTALLET IALT ER          ",
0120 PRINT USING "#####": TIMERIALT
0130 PRINT "UDBETALT LØN IALT UDGØR KR.  ",
0140 PRINT USING "#####.##": LØNIALT
0150 GOTO 0030
```

Der er en ny slags programsætning, i hvilken linje?

Oversæt hvad linjen betyder:

GOTO

GOTO-SÆTNINGEN BEVIRKER, AT MASKINEN SPRINGER TIL DEN LINJE, DER STÅR UMIDDELBART EFTER GOTO OG FORTSÆTTER DERFRA.

Skriv herunder hvad programmet udfører:

TÆLLEFIDUS

Hvis vi ønsker, at programmet i øvelse 4.43 kun skal udføres et bestemt antal gange f.eks. 3, kan vi benytte en »TÆLLEFIDUS« og en IF-THEN sætning.

Øvelse 4.44

Læs nedenstående program grundigt igennem.

```
0010 TÆLLEFIDUS:=0
0020 TIERIALT:=0
0030 LØNIALT:=0
0040 PRINT "INDTAST ANTAL TIMER"
0050 INPUT TIMEANTAL
0060 PRINT "INDTAST TIMELØNNEN"
0070 INPUT TIMELØN
0080 PRINT
0090 TIERIALT:=TIERIALT+TIMEANTAL
0100 DAGLØN:=TIMEANTAL*TIMELØN
0110 LØNIALT:=LØNIALT+DAGLØN
0120 TÆLLEFIDUS:=TÆLLEFIDUS+1
0130 IF TÆLLEFIDUS<3 THEN
0140 GOTO 0040
0150 ENDIF
0160 PRINT "TIMEANTALLET IALT ER"
0170 PRINT USING "#####": TIERIALT
0180 PRINT "UDBETALT LØN IALT UDGØR KR."
0190 PRINT USING "#####.##": LØNIALT
```

Hvad udfører »TÆLLEFIDUSEN«?

Du kan afprøve programmet ved at taste LOAD ØVELSE20.

Øvelse 4.45

Ret programmet, så det gennemløbes 5 gange.

Afprøv om dit program virker, som det skal.

Øvelse 4.46

Skriv et program, hvor du indtaster:

KAPITAL
RENTESATS

Programmet skal give nedenstående udskrift:

RENTEN UDGØR KR. XXXXX.XX OM ÅRET

Formlen for renteberegning er:

$$\frac{\text{KAPITAL} \times \text{RENTESATS}}{100}$$

Øvelse 4.47

Udvid det ovenstående program, så det kan beregne renten af en kapital i et bestemt antal dage.

INDDATA:

KAPITAL
RENTESATS
ANTAL DAGE

Programmet skal give nedenstående udskrift:

RENTEN UDGØR KR. XXXXX.XX FOR DE XXX DAGE

Formlen for beregningen er:

$$\frac{\text{KAPITAL} \times \text{RENTESATS} \times \text{DAGEANTAL}}{100 \times 360}$$

Øvelse 4.48

Skriv et program, der omregner danske kroner til svenske kroner.

INDDATA:

BELØB I DANSKE KRONER

UDDATA:

BELØB I DANSKE KR. XXXXX.XX
BELØB I SVENSKER KR. XXXXX.XX

Formlen for omregning af danske kroner til fremmed mønt er:

$$\frac{\text{DANSKE KR.} \times 100}{\text{KURS}}$$

Kursen på svenske kroner kan du finde i kurslisten.

Øvelse 4.49

Udvid det ovenstående program, så du kan indtaste kurs.

Prøv programmet med forskellige møntsorter; kurserne får du fra nedenstående kursliste.

VALUTA- NOTERING	
Valutakurser	
København 23/3 1981 (Reuter)	
US-dollar	657,35 (651,50)
pund sterling	1475,20 (1478,35)
D-mark	314,85 (314,85)
svenske kroner	143,47 (143,28)
norske kroner	122,45 (122,05)
fr. franc, varer	133,65 (133,60)
belg. franc	19,215 (19,22)
schweizerfranc	345,90 (345,65)
gylden	284,65 (284,80)
lire	0,6390 (0,6467)
finmark	163,01 (162,60)
islandske kroner	102,16 (102,50)
østrigske schilling	44,53 (44,55)
pesetas	7,77 (7,775)
escudos	11,71 (11,71)
canadiske dollar	55,75 (550,55)
japanske yen	3,1570 (3,1235)
irske pund	1149,50 (1151,00)

Øvelse 4.50

Skriv et program, der omregner svenske kroner til danske kroner.

INDDATA:

BELØB I SVENSKE KRONER

UDDATA:

```
BELØB I SVENSKE KR.      XXXXX.XX
BELØB I DANSKE KR.      XXXXX.XX
```

Formlen for omregning af fremmed mønt til danske kroner er:

$$\frac{\text{DANSKE KR.} \times \text{KURS}}{100}$$

Øvelse 4.51

Udvid det ovenstående program, så du kan indtaste kurs.

Prøv programmet med forskellige møntsorter; kurserne får du fra kurslisten.

Øvelse 4.52

Skriv et program, der beregner, hvor stor en kapital vil blive år for år, når den forrentes med 12% p.a. (pro anno).

Programmet skal vise, hvor meget renten er år for år.

Programmet skal vise resultaterne for en periode på 5 år.

INDDATA:

KAPITAL

UDDATA:

```
KAPITAL KR.      XXXXX.XX
RENTESATS              XX
```

```
EFTER X AR UDGØR KAPITALEN KR. XXXXX.XX RENTEN KR. XXXX.XX
EFTER X AR UDGØR KAPITALEN KR. XXXXX.XX RENTEN KR. XXXX.XX
EFTER X AR UDGØR KAPITALEN KR. XXXXX.XX RENTEN KR. XXXX.XX
EFTER X AR UDGØR KAPITALEN KR. XXXXX.XX RENTEN KR. XXXX.XX
EFTER X AR UDGØR KAPITALEN KR. XXXXX.XX RENTEN KR. XXXX.XX
```

Øvelse 4.53

Ret dit program, så der kan indtastes variabel rente og antal år.

Programmer til øvelserne 4.41 til 4.53 er lagret under navnene ØVELSE17 til ØVELSE29.

Du kan hente dem frem ved at taste LOAD og det pågældende ØVELSENr.

Efterskrift

Der gøres opmærksom på, at programmeringsafsnittet følges op af opgaver bagest i den endelige bog.

